

# 「FA教育システム」の実践研究Ⅱ

工業科 工藤雄司・茂木好和・金城幸廣  
深澤孝之・小高昌人・大平典男

## 1. はじめに

FAシステムは、各種NC工作機械や産業用ロボット、自動倉庫の間を無人搬送車等で結び、部品・材料が自動的に移動し、各種製品を混合して生産するシステムである。このような状況の産業界に卒業生を送り出している本校では、メカトロニクス教育の推進・発展を目指した実践を行ってきた。

この授業実践では、生徒のFAシステムに対する理解を効率的に深め、選択制の授業形態にも容易に対応できるよう開発した「FA教育システム」における授業実践を報告する。

## 2. 「FA教育システム」の特徴

自動倉庫セル、加工セル、組立セル、無人搬送車があり、各々を制御するパソコンがネットワークで結ばれている。そして、各セルの作業内容・進行状況等がネットワークを通じて送受信されるので、ホストパソコンは、ネットワークを監視することにより、システム全体を高機能に制御する。

また、自動倉庫セルの直交座標型ロボット、加工セルの垂直多関節型ロボット、組立セルの水平多関節型ロボットを学習することにより、産業用ロボットの主要な方式のロボットを全て学習できる。したがって、生徒のFAシステムに対する理解を効率的に深めることができるシステムである。

## 3. 本システムを利用する科目と内容

(1) 電子機械実習Ⅰ：全システムをネットワークにより連動させた全自動運転を行い、FAシステムの概要を理解する。

(2) 電子機械実習Ⅱ：各セル毎に、セル単位のI/O入出力によるコンピュータ制御実習を行い、コンピュータ制御システムを理解する。自動倉庫セルの制御プログラムの例を資料Ⅰに示す。

各セルの動作を、ネットワークを通して制御し、システムを連動させる技術を理解する。

各種ロボットのティーチングや、ロボット制御命令を理解し、FAシステムにおける各種ロボットの制御方法

を理解する。

(3) プログラミング技術Ⅱ：C言語による制御プログラミング実習を行い、コンピュータ制御システムのプログラミング技術を習得する。

## 4. 各セルの使用方法

各セルを使用するに当たって、各セル毎に起動方法が異なる。そして、ロボットの動作を、予めティーチングBOXで入力しておき、コマンドで指定する。

(1) 自動倉庫セルの起動方法とロボットコマンド

①非常停止ボタンの解除

②制御盤扉の緑の運転準備ボタンを押し、ランプが点灯することを確認する。

③コンベア上のパレットを取り除き、倉庫ラックのパレットを全て正しく入れる。

④ロボットを原点復帰するため、ティーチングBOXの

1) F1 (START) キーを押し、

2) 再度F1 (START) キーを押し、

3) F2 (PLAY) キーを押し、

4) リターンキーを押し、

5) F2 (GO) キーを押す。

⑤プログラム[SOUKO]を実行する。

ロボットコマンド表

動作	第1コマンド	第2コマンド
ラック・チェック	2 3	目的ラック(1~16)
ラック→コンベア	1 2	目的ラック(1~16)
コンベア→ラック	1 3	目的ラック(1~16)

(2) 加工セルの起動方法とロボットコマンド

①非常停止ボタンの解除

②倉庫セルの制御盤扉の緑の運転準備ボタンを押し、ランプが点灯することを確認する。

③コンベア出入口にパレット・ワークをセットし、供給ユニットのワークを取り除く。

④ロボットを原点復帰・運転するため、プログラム

[VRINI] を実行する。

⑤プログラム [KAKOU] を実行する。

ロボットコマンド表

動 作	第1コマンド	第2コマンド
コンベア→供給	1 0	1 4
供給→コンベア	1 0	1 5
コンベア→NG	1 0	1 3

(3) 組立セルの起動方法とロボットコマンド

①非常停止ボタンの解除

②倉庫セルの制御盤扉の緑の運転準備ボタンを押し、ランプが点灯することを確認する。

③コンベア出入口にパレット・サイコロの1穴をセットし、トレイに部品をセットする。

④ロボットを原点復帰・運転するため、プログラム [HRINI] を実行する。

⑤プログラム [KUMITA] を実行する。

ロボットコマンド表

動 作	第1コマンド	第2コマンド
トレイ→サイコロ	3 (1列目)	目的トレ-(1~16)
トレイ→サイコロ	5 (2列目)	目的トレ-(1~16)
サイコロ→トレイ	4 (1列目)	目的トレ-(1~16)
サイコロ→トレイ	6 (2列目)	目的トレ-(1~16)
サイコロ→NG	1 0	2 6

## 5. 各セルの入出力ポートの仕様

BASIC言語やC言語のプログラムで様々な制御を

行う場合、パソコンの入出力ポートのアドレスと、各々のポートに接続されている信号の種類や機器との対応表が必要となるので、資料2に示す。各々のポートに対する入出力を行うだけで、リレーやソレノイドの制御も、各種ロボットの制御においても、所定の動作を行う。

## 6. おわりに

本実践は2年目に入り、生徒の学習意欲は大いに喚起されている状態である。連動運転と、個別運転の切り替えも容易であり、選択制の授業形態に対応できているので、「FA教育システム」としては効果的なシステムと考えられる。

## 参考文献

- 1) 亀山・工藤他, 1991, 工業科における「課題研究」の実践的研究, 筑波大学校教育部紀要, 第14巻, PP117-136
- 2) 深作・工藤他, 1992, 情報技術教育の実践研究Ⅰ, 筑波大学校教育部紀要, 第15巻, PP47-65
- 3) 工藤・深作他, 1992, コンピュータ教育の実践研究Ⅰ, 筑波大附属坂戸高紀要, 第31集, PP47-51
- 4) 工藤・深作他, 1993, コンピュータ教育の実践研究Ⅱ, 筑波大附属坂戸高紀要, 第32集, PP89-94
- 5) 工藤・深作他, 1996, コンピュータ教育の実践研究Ⅲ, 筑波大附属坂戸高紀要, 第33・34集, PP73-75
- 6) 工藤・深作他, 1997, 「FA教育システム」の実践研究Ⅱ, 筑波大附属坂戸高紀要, 第35集, PP113-116

## 資料1 自動倉庫セルの制御プログラムの例

```

1000 'SAVE "SOUKO".A
1010 SCREEN 3,0,0,1 : CONSOLE 0,25,1,1 : CLS 3      '画面初期化
1020 ON STOP GOSUB *P.END : STOP ON                 '[STOP]を押すと *P.END に飛ぶ
1030 OUT &HD4,4                                       'ストッパー解除
1040
1050 ' 画面に文字を書く (LOCATE X,Y 位置に, PRINT "文字" を書く)
1060 LOCATE 5,3: PRINT "倉庫ラック 1:□ 2:□ 3:□ 4:□"
1070 LOCATE 5,4: PRINT "倉庫ラック 5:□ 6:□ 7:□ 8:□"
1080 LOCATE 5,5: PRINT "倉庫ラック 9:□ 10:□ 11:□ 12:□"
1090 LOCATE 5,6: PRINT "倉庫ラック 13:□ 14:□ 15:□ 16:□ コンベア:□"
1100
1110 LOCATE 5,10: PRINT "      128 64 32 16 8 4 2 1 各ビットの10進数値"
1120 LOCATE 5,11: PRINT "入力ポート: 7 6 5 4 3 2 1 0 ビット"
1130 LOCATE 5,12: PRINT "&HD4: ○ ○ ○ ○ ○ ○ ○ ○"
1140 LOCATE 5,13: PRINT "&HD5: ○ ○ ○ ○ ○ ○ ○ ○"
1150 LOCATE 5,14: PRINT "&HD6: ○ ○ ○ ○ ○ ○ ○ ○"
1160 LOCATE 5,15: PRINT "&HD7: ○ ○ ○ ○ ○ ○ ○ ○"
1170

```

```

1180 LOCATE 5,17: PRINT "出力ポート: 7 6 5 4 3 2 1 0 ビット"
1190 LOCATE 5,18: PRINT "&HD7: ○ ○ ○ ○ ○ ○ ○ ○"
1200
1210 ' ロボット初期化
1220 OD= 128 : GOSUB *R.OUT : OD= 0 : GOSUB *R.OUT ' *R.OUT に飛び、
1230 OD= 128 : GOSUB *R.OUT : OD= 0 : GOSUB *R.OUT ' RETURN で戻ってくる
1240
1250 *MAIN ' メイン プログラム *MAIN はここから始まる
1260 LOCATE 5,1: PRINT "倉庫ラックに入庫なら[1], 出庫なら[0]を押して下さい。"
1270 GOSUB *P.IN ' *P.IN に飛び、RETURN で戻ってくる
1280 FOR R = 1 TO 16 ' 変数 R に 1 を入れ、NEXT R までの命令を実行する
1290 AS=INKEY$ ' 文字変数 AS にキーボードから1文字入れる、押さないと空入力
1300 IF AS="1" THEN *W.IN ' 文字変数 AS が 1 のときは、*W.IN に飛び
1310 IF AS="0" THEN *W.OUT ' 文字変数 AS が 0 のときは、*W.OUT に飛び
1320 GOSUB *R.CHK ' *R.CHK に飛び、RETURN で戻ってくる
1330 NEXT R ' FOR に戻り、R に 2 を入れ繰り返し、16 を繰り返すと終了
1340 GOSUB *CV.CHK ' *CV.CHK に飛び、RETURN で戻ってくる
1350 GOTO *MAIN ' *MAIN に飛び
1360
1370 *W.IN ' 入庫プログラム
1380 LOCATE 5,1: PRINT "倉庫ラックの何番に入庫しますか"
1390 LOCATE 35,1: INPUT R ' ? を出し、変数 R に入力
1400 IF R<1 OR R>16 THEN *W.IN ' 1 から 16 以外の入力はやリ直し
1410 GOSUB *R.CHK : IF RD=4 THEN *R.ER1 ' &HD7=4(有)ならば、R.ER1に飛び
1420 GOSUB *CV.CHK : IF D=0 THEN *C.ER1 ' &HD4=0(無)ならば、C.ER1に飛び
1430 COM1=13 : COM2=R : GOSUB *R.RUN ' ロボット第1コマンド=13, 第2コマンド=R
1440 GOSUB *R.CHK : GOSUB *CV.CHK ' ラックチェック, コンベアチェック
1450 GOTO *MAIN
1460
1470 *W.OUT ' 出庫プログラム
1480 LOCATE 5,1: PRINT "倉庫ラック何番から出庫しますか"
1490 LOCATE 35,1: INPUT R ' ? を出し、変数 R に入力
1500 IF R<1 OR R>16 THEN *W.OUT ' 1 から 16 以外の入力はやリ直し
1510 GOSUB *R.CHK : IF RD=2 THEN *R.ER2 ' &HD7=2(無)ならば、R.ER2に飛び
1520 GOSUB *CV.CHK : IF D=1 THEN *C.ER2 ' &HD4=1(有)ならば、C.ER2に飛び
1530 COM1=12 : COM2=R : GOSUB *R.RUN ' ロボット第1コマンド=13, 第2コマンド=R
1540 GOSUB *R.CHK : GOSUB *CV.CHK ' ラックチェック, コンベアチェック
1550 GOTO *MAIN
1560
1570 *R.ER1
1580 COLOR 6 : LOCATE 5,1 : PRINT R:"番ラックは入庫済みです！" : COLOR 7
1590 FOR I=0 TO 100000! : NEXT I
1600 GOTO *MAIN
1610
1620 *R.ER2
1630 COLOR 6 : LOCATE 5,1 : PRINT R:"番ラックは出庫済みです！" : COLOR 7
1640 FOR I=0 TO 100000! : NEXT I
1650 GOTO *MAIN
1660
1670 *C.ER1
1680 COLOR 6: LOCATE 5,1: PRINT "コンベア上にパレットがありません！" : COLOR 7
1690 FOR I=0 TO 100000! : NEXT I
1700 GOTO *MAIN
1710
1720 *C.ER2
1730 COLOR 6: LOCATE 5,1: PRINT "コンベア上にパレットがあります！" : COLOR 7
1740 FOR I=0 TO 100000! : NEXT I
1750 GOTO *MAIN
1760
1770 *R.CHK ' 倉庫ラックパレット確認
1780 LOCATE 50,18 : PRINT R:"番ラック確認"
1790 IF R>= 1 AND R<= 4 THEN X= R *7+11 :Y=3 :GOSUB *R.CHK2 'X=18,25,32,39
1800 IF R>= 5 AND R<= 8 THEN X=(R- 4)*7+11 :Y=4 :GOSUB *R.CHK2 'X=18,25,32,39
1810 IF R>= 9 AND R<=12 THEN X=(R- 8)*7+11 :Y=5 :GOSUB *R.CHK2 'X=18,25,32,39
1820 IF R>=13 AND R<=16 THEN X=(R-12)*7+11 :Y=6 :GOSUB *R.CHK2 'X=18,25,32,39

```

```

1830 RETURN      ' ↑ X を求めるため          ↑ Y を求めるため
1840 '
1850 *R.CHK2
1860 OD=23+128 : GOSUB *R.OUT : OD= 0 : GOSUB *R.OUT '第1 コマンド (ラックチェック)
1870 OD=R +128 : GOSUB *R.OUT : OD= 0 : GOSUB *R.OUT '第2 コマンド
1880 OD= 128 : GOSUB *R.OUT '実行コマンドの初め
1890 '
1900 FOR I=0 TO 5000 : NEXT I
1910 GOSUB *P.IN : RD=D7 AND 6 ' &HD7 から ビット1,2 だけ取り出す RD=0,2,4,6
1920 COLOR 6 : LOCATE X,Y : IF RD=4 THEN PRINT "■" ELSE PRINT "□" 'X,Y 使用
1930 COLOR 7 : OD= 0 : GOSUB *R.OUT '実行コマンドの残り
1940 RETURN
1950 '
1960 *CV.CHK ' コンベア上パレット確認
1970 LOCATE 50,18 : PRINT "コンベア確認 "
1980 GOSUB *P.IN : D=D4 AND 1 ' &HD4 から ビット0 だけ取り出す D=0,1
1990 COLOR 6 : LOCATE 52,6 : IF D=1 THEN PRINT "■" ELSE PRINT "□"
2000 COLOR 7 : FOR I=0 TO 30000 : NEXT I
2010 RETURN
2020 '
2030 *R.RUN ' ロボット作動コマンド
2040 LOCATE 50,18 : PRINT "ロボット作動中 "
2050 OD=COM1+128 : GOSUB *R.OUT : OD= 0 : GOSUB *R.OUT '第1 コマンド
2060 OD=COM2+128 : GOSUB *R.OUT : OD= 0 : GOSUB *R.OUT '第2 コマンド
2070 OD= 128 : GOSUB *R.OUT : OD= 0 : GOSUB *R.OUT '実行コマンド
2080 '
2090 *R.RUN2 ' &HD7 から ビット7 だけ取り出し,D=128 のロボット動作中は繰り返す
2100 GOSUB *P.IN : D=D7 AND 128 : IF D=128 THEN *R.RUN2
2110 LOCATE 50,18 : PRINT "ロボット作動終了"
2120 RETURN
2130 '
2140 *R.OUT
2150 OUT &HD7,OD : L=18 : D=OD : C=4 : GOSUB *GRAP ' &HD7ポートに変数OD を出力
2160 FOR I=0 TO 10 : GOSUB *P.IN : NEXT I 'ポート入力で時間待ち
2170 RETURN
2180 '
2190 *P.IN
2200 D4=INP(&HD4) : L=12 : D=D4 : C=2 : GOSUB *GRAP ' &HD4ポートから変数D4に入力
2210 D5=INP(&HD5) : L=13 : D=D5 : C=2 : GOSUB *GRAP ' &HD5 " D5 "
2220 D6=INP(&HD6) : L=14 : D=D6 : C=2 : GOSUB *GRAP ' &HD6 " D6 "
2230 D7=INP(&HD7) : L=15 : D=D7 : C=2 : GOSUB *GRAP ' &HD7 " D7 "
2240 RETURN
2250 '
2260 *GRAP
2270 COLOR C ' C = 0:黒 1:青 2:赤 3:薄紫 4:緑 5:水色 6:黄色 7:白
2280 LOCATE 16,L : IF D AND 128 THEN PRINT"●" ELSE PRINT"○"
2290 LOCATE 19,L : IF D AND 64 THEN PRINT"●" ELSE PRINT"○"
2300 LOCATE 22,L : IF D AND 32 THEN PRINT"●" ELSE PRINT"○"
2310 LOCATE 25,L : IF D AND 16 THEN PRINT"●" ELSE PRINT"○"
2320 LOCATE 28,L : IF D AND 8 THEN PRINT"●" ELSE PRINT"○"
2330 LOCATE 31,L : IF D AND 4 THEN PRINT"●" ELSE PRINT"○"
2340 LOCATE 34,L : IF D AND 2 THEN PRINT"●" ELSE PRINT"○"
2350 LOCATE 37,L : IF D AND 1 THEN PRINT"●" ELSE PRINT"○"
2360 COLOR 7
2370 RETURN
2380 '
2390 *P.END
2400 OUT &HD4,0
2410 CLOSE : COLOR 7 : CLS 3
2420 END

```

# 自動倉庫セル 入力ポート仕様

筑波大学附属坂戸高等学校

ポート/ビット	I/O	HEX	信号	機器	シーケンス	
&H D 4	IN 0	01	01	パレット到着検出	近接センサー	000
	IN 1	02	02	ワーク有無検出	透過型センサー	001
	IN 2	03	04	パレット通過検出	透過型センサー	002
	IN 3	04	08	ストッパー解除確認	オートスイッチ	003
	IN 4	05	10			004
	IN 5	06	20			005
	IN 6	07	40			006
	IN 7	08	80	セル非常停止		007
&H D 5	IN 0	09	01	パレットNoビット検出 (8)	近接センサー	008
	IN 1	10	02	パレットNoビット検出 (4)	近接センサー	009
	IN 2	11	04	パレットNoビット検出 (2)	近接センサー	010
	IN 3	12	08	パレットNoビット検出 (1)	近接センサー	011
	IN 4	13	10	原点位置確認 (2)		012
	IN 5	14	20	原点位置確認 (1)		013
	IN 6	15	40	ロボット位置確認 (2)		014
	IN 7	16	80	ロボット位置確認 (1)		015
&H D 6	IN 0	17	01	AGV着 (パレット無し)	光I/O	100
	IN 1	18	02	AGV着 (パレット有り)	光I/O	101
	IN 2	19	04	AGVワーク引き込み完了	光I/O	102
	IN 3	20	08	NGワーク	光I/O	103
	IN 4	21	10			104
	IN 5	22	20			105
	IN 6	23	40	AGV光I/O Ready	光I/O	106
	IN 7	24	80	AGV確認信号 (ハンドシェイク)	光I/O	107
&H D 7	IN 0	25	01	ハンドシェイク信号	ロボットコントローラ (THK)	108 (302R)
	IN 1	26	02	ロボットエラー1 (ラックパレット無し)	ロボットコントローラ (THK)	109 (303R)
	IN 2	27	04	ロボットエラー2 (ラックパレット有り)	ロボットコントローラ (THK)	110 (304R)
	IN 3	28	08	予備	ロボットコントローラ (THK)	111 (305R)
	IN 4	29	10	予備	ロボットコントローラ (THK)	112 (306R)
	IN 5	30	20	予備	ロボットコントローラ (THK)	113 (307R)
	IN 6	31	40	プログラム実行中	ロボットコントローラ (THK)	114 (308R)
	IN 7	32	80	ロボット動作中	ロボットコントローラ (THK)	115 (309R)

# 自動倉庫セル 出力ポート仕様

筑波大学附属坂戸高等学校

ポート/ビット	I/O	HEX	信号	機器	シーケンス	
&H D 4	OUT 0	01	01	コンベア駆動 (引き込み方向)	リレー	10000
	OUT 1	02	02	コンベア駆動 (排出方向)	リレー	10001
	OUT 2	03	04	ストッパー解除	シングルソレノイド	10002
	OUT 3	04	08			10003
	OUT 4	05	10			10004
	OUT 5	06	20			10005
	OUT 6	07	40			10006
	OUT 7	08	80			10007
&H D 5	OUT 0	09	01	AGVコースパターン (1)	光 I/O	10008
	OUT 1	10	02	AGVコースパターン (2)	光 I/O	10009
	OUT 2	11	04	AGVコースパターン (4)	光 I/O	10010
	OUT 3	12	08	AGVコースパターンストロブ	光 I/O	10011
	OUT 4	13	10	AGVコンベア動作	光 I/O	10012
	OUT 5	14	20	AGVスタート	光 I/O	10013
	OUT 6	15	40	NGワーク	光 I/O	10014
	OUT 7	16	80	AGV確認信号 (ストロブ)	光 I/O	10015
&H D 6	OUT 0	17	01			10100
	OUT 1	18	02			10101
	OUT 2	19	04			10102
	OUT 3	20	08			10103
	OUT 4	21	10			10104
	OUT 5	22	20			10105
	OUT 6	23	40			10106
	OUT 7	24	80			10107
&H D 7	OUT 0	25	01	ロボットコマンド (1)	ロボットコントローラ (THK)	10108
	OUT 1	26	02	ロボットコマンド (2)	ロボットコントローラ (THK)	10109
	OUT 2	27	04	ロボットコマンド (4)	ロボットコントローラ (THK)	10110
	OUT 3	28	08	ロボットコマンド (8)	ロボットコントローラ (THK)	10111
	OUT 4	29	10	ロボットコマンド (16)	ロボットコントローラ (THK)	10112
	OUT 5	30	20	予備	ロボットコントローラ (THK)	10113
	OUT 6	31	40	予備	ロボットコントローラ (THK)	10114
	OUT 7	32	80	ストロブ信号	ロボットコントローラ (THK)	10115

# 加工セル 入力ポート仕様

筑波大学附属坂戸高等学校

ポート/ビット	I/O	HEX	信号	機器	シーケンス
&H D 0	IN 0	01	形状識別 (1)	光ファイバー	000
	IN 1	02	形状識別 (2)	光ファイバー	001
	IN 2	03	形状識別 (3)	光ファイバー	002
	IN 3	04	形状識別 (4)	光ファイバー	003
	IN 4	05	形状識別 (5)	光ファイバー	004
	IN 5	06	形状識別 (6)	光ファイバー	005
	IN 6	07	形状識別 (7)	光ファイバー	006
	IN 7	08	形状識別 (8)	光ファイバー	007
&H D 1	IN 0	09	形状識別 (9)	光ファイバー	008
	IN 1	10	1 STパレット到着検出	近接センサ	009
	IN 2	11	パレットNoビット検出 (8)	近接センサ	010
	IN 3	12	パレットNoビット検出 (4)	近接センサ	011
	IN 4	13	パレットNoビット検出 (2)	近接センサ	012
	IN 5	14			
	IN 6	15			
	IN 7	16	セル非常停止		013
&H D 2	IN 0	17	パレットNoビット検出 (1)	近接センサ	100
	IN 1	18	2 STパレット到着検出	反射型フォトセンサ	101
	IN 2	19	1 ST位置決め完了検出	オートスイッチ	102
	IN 3	20	2 ST位置決め完了検出	オートスイッチ	103
	IN 4	21	1 STストッパー解除確認	オートスイッチ	104
	IN 5	22	パレット通過検出	透過型センサ	105
	IN 6	23	加工終了確認		106
	IN 7	24			
&H D 3	IN 0	25	原点位置確認 (2)		
	IN 1	26	原点位置確認 (1)		
	IN 2	27	ロボット位置確認 (2)		
	IN 3	28	ロボット位置確認 (1)		
	IN 4	29	ワーク排出確認	オートスイッチ	107
	IN 5	30	ワーク供給確認	オートスイッチ	108
	IN 6	31	供給側ワーク有無検出	透過型センサ	109
	IN 7	32	1 STワーク有無検出	透過型センサ	110
&H D 4	IN 0	33	AGV着 (パレット無し)	光 I/O	
	IN 1	34	AGV着 (パレット有り)	光 I/O	
	IN 2	35	AGVワーク引き込み完了	光 I/O	
	IN 3	36	NGワーク	光 I/O	
	IN 4	37	予備	光 I/O	
	IN 5	38	予備	光 I/O	
	IN 6	39	AGV光 I/O Ready	光 I/O	
	IN 7	40	AGV確認信号 (ハンドシェイク)	光 I/O	
&H D 5	IN 0	41	ハンドシェイク信号	ロボットコントローラ (RV-M1)	111
	IN 1	42	ロボットエラー 1	ロボットコントローラ (RV-M1)	112
	IN 2	43	ロボットエラー 2	ロボットコントローラ (RV-M1)	113
	IN 3	44	予備	ロボットコントローラ (RV-M1)	
	IN 4	45	予備	ロボットコントローラ (RV-M1)	
	IN 5	46	予備	ロボットコントローラ (RV-M1)	
	IN 6	47	予備	ロボットコントローラ (RV-M1)	
	IN 7	48	ロボット動作中	ロボットコントローラ (RV-M1)	114

# 加工セル 出力ポート仕様

筑波大学附属坂戸高等学校

ポート/ビット	I/O	HEX	信号	機器	シーケンス
&H D 8	OUT 0	01	コンベア駆動 (引き込み方向)	リレー	10000
	OUT 1	02	コンベア駆動 (排出方向)	リレー	10001
	OUT 2	03	1 STストッパー解除	シングルソレノイド	10002
	OUT 3	04	1 ST位置決めON	シングルソレノイド	10003
	OUT 4	05	2 ST位置決めON	シングルソレノイド	10004
	OUT 5	06	ワーク供給ユニット (供給)	ダブルソレノイド	10005
	OUT 6	07	ワーク供給ユニット (排出)	ダブルソレノイド	10006
	OUT 7	08			10007
&H D 9	OUT 0	09	AGVコースパターン (1)	光 I/O	10008
	OUT 1	10	AGVコースパターン (2)	光 I/O	10009
	OUT 2	11	AGVコースパターン (4)	光 I/O	10010
	OUT 3	12	AGVコースパターンストロブ	光 I/O	10011
	OUT 4	13	AGVコンベア動作	光 I/O	10012
	OUT 5	14	AGVスタート	光 I/O	10013
	OUT 6	15	NGワーク	光 I/O	10014
	OUT 7	16	AGV確認信号 (ストロブ)	光 I/O	10015
&H D A	OUT 0	17			10100
	OUT 1	18			10101
	OUT 2	19			10102
	OUT 3	20			10103
	OUT 4	21			10104
	OUT 5	22			10105
	OUT 6	23			10106
	OUT 7	24			10107
&H D B	OUT 0	25	ロボットコマンド (1)	ロボットコントローラ (RV-M1)	10108 (OUT0R)
	OUT 1	26	ロボットコマンド (2)	ロボットコントローラ (RV-M1)	10109 (OUT1R)
	OUT 2	27	ロボットコマンド (4)	ロボットコントローラ (RV-M1)	10110 (OUT2R)
	OUT 3	28	ロボットコマンド (8)	ロボットコントローラ (RV-M1)	10111 (OUT3R)
	OUT 4	29	ロボットコマンド (16)	ロボットコントローラ (RV-M1)	10112 (OUT4R)
	OUT 5	30	ストロブ信号	ロボットコントローラ (RV-M1)	10113 (OUT5R)
	OUT 6	31	予備	ロボットコントローラ (RV-M1)	10114 (OUT6R)
	OUT 7	32	予備	ロボットコントローラ (RV-M1)	10115 (OUT7R)

## 組立セル 入力ポート仕様

筑波大学附属坂戸高等学校

ポート/ビット	I/O	HEX	信号	機器	シーケンス
&H D 4	IN 0	01 01	1 S Tパレット到着検出	近接センサ	000
	IN 1	02 02	2 S Tパレット到着検出	近接センサ	001
	IN 2	03 04	ワーク有無検出	透過型センサ	002
	IN 3	04 08	パレット通過検出	透過型センサ	003
	IN 4	05 10	1 S T位置決め完了	オートスイッチ	004
	IN 5	06 20	2 S T位置決め完了	オートスイッチ	005
	IN 6	07 40	1 S Tストッパー解除確認	オートスイッチ	006
	IN 7	08 80	セル非常停止		007
&H D 5	IN 0	09 01	パレットNoビット検出 (1)	近接センサ	008
	IN 1	10 02	パレットNoビット検出 (2)	近接センサ	009
	IN 2	11 04	パレットNoビット検出 (4)	近接センサ	010
	IN 3	12 08	パレットNoビット検出 (8)	近接センサ	011
	IN 4	13 10	原点位置確認 (1)	透過型センサ	012
	IN 5	14 20	原点位置確認 (2)	透過型センサ	013
	IN 6	15 40	ロボット位置確認 (1)	反射型フォト	014
	IN 7	16 80	ロボット位置確認 (2)	反射型フォト	015
&H D 6	IN 0	17 01	AGV着 (パレット無し)	光 I/O	100
	IN 1	18 02	AGV着 (パレット有り)	光 I/O	101
	IN 2	19 04	AGVワーク引き込み完了	光 I/O	102
	IN 3	20 08	NGワーク	光 I/O	103
	IN 4	21 10			104
	IN 5	22 20			105
	IN 6	23 40	AGV光 I/O Ready	光 I/O	106
	IN 7	24 80	AGV確認信号 (ハンドシェイク)	光 I/O	107
&H D 7	IN 0	25 01	ハンドシェイク信号	ロボットコントローラ (RH-M2)	108
	IN 1	26 02	ロボットエラー 1	ロボットコントローラ (RH-M2)	109
	IN 2	27 04	ロボットエラー 2	ロボットコントローラ (RH-M2)	110
	IN 3	28 08	予備	ロボットコントローラ (RH-M2)	111
	IN 4	29 10	予備	ロボットコントローラ (RH-M2)	112
	IN 5	30 20	予備	ロボットコントローラ (RH-M2)	113
	IN 6	31 40	プログラム実行中	ロボットコントローラ (RH-M2)	114
	IN 7	32 80	ロボット動作中	ロボットコントローラ (RH-M2)	115

## 組立セル 出力ポート仕様

筑波大学附属坂戸高等学校

ポート/ビット	I/O	HEX	信号	機器	シーケンス
&H D 4	OUT 0	01 01	コンベア駆動 (引き込み方向)	リレー	10000
	OUT 1	02 02	コンベア駆動 (排出方向)	リレー	10001
	OUT 2	03 04	1 S Tストッパー解除	シングルソレノイド	10002
	OUT 3	04 08	1 S T位置決め	シングルソレノイド	10003
	OUT 4	05 10	2 S T位置決め	シングルソレノイド	10004
	OUT 5	06 20			10005
	OUT 6	07 40			10006
	OUT 7	08 80			10007
&H D 5	OUT 0	09 01	AGVコースパターン (1)	光 I/O	10008
	OUT 1	10 02	AGVコースパターン (2)	光 I/O	10009
	OUT 2	11 04	AGVコースパターン (4)	光 I/O	10010
	OUT 3	12 08	AGVコースパターンストローブ	光 I/O	10011
	OUT 4	13 10	AGVコンベア動作	光 I/O	10012
	OUT 5	14 20	AGVスタート	光 I/O	10013
	OUT 6	15 40	NGワーク	光 I/O	10014
	OUT 7	16 80	AGV確認信号 (ストローブ)	光 I/O	10015
&H D 6	OUT 0	17 01			10100
	OUT 1	18 02			10101
	OUT 2	19 04			10102
	OUT 3	20 08			10103
	OUT 4	21 10			10104
	OUT 5	22 20			10105
	OUT 6	23 40			10106
	OUT 7	24 80			10107
&H D 7	OUT 0	25 01	ロボットコマンド (1)	ロボットコントローラ (RH-M2)	10108
	OUT 1	26 02	ロボットコマンド (2)	ロボットコントローラ (RH-M2)	10109
	OUT 2	27 04	ロボットコマンド (4)	ロボットコントローラ (RH-M2)	10110
	OUT 3	28 08	ロボットコマンド (8)	ロボットコントローラ (RH-M2)	10111
	OUT 4	29 10	ロボットコマンド (16)	ロボットコントローラ (RH-M2)	10112
	OUT 5	30 20	ストローブ信号	ロボットコントローラ (RH-M2)	10113
	OUT 6	31 40	予備	ロボットコントローラ (RH-M2)	10114
	OUT 7	32 80	予備	ロボットコントローラ (RH-M2)	10115